Proceedings of the 1984
Winter Simulation Conference
S. Sheppard, U. Pooch, D. Pegden (eds.)

CONTINUOUS SYSTEM SIMULATION LANGUAGES (CSSL'S)

Dr. Ralph C. Huntsinger
Computer Science and Mechanical Engineering Departments
California State University, Chico
Chico, California 95929

This tutorial introduces CSSL's (Continuous System Simulation Languages) by using examples from three of the popular commercial languages used in North America at the present time. The languages are CSSL-IV (1), DSL/VS (2) and ISIM (3).

Continuous System Simulation Languages are user-oriented software systems. CSSL's are designed to assist engineers and scientists to mathematically model, analyze, and evaluate the dynamic behavior of physical phenomena. By providing a set of tools for computer-aided-analysis, they make it easy for the user to get his simulation on the computer quickly and to easily conduct experiments, collect data and present that data in useful form with minimal knowledge of the computer system itself.

CSSL's are easily learned and applied to many types of problems in all sciences and engineering disciplines. The problems can usually be coded in a short time, executed immediately and evaluated quickly by inspection of graphic output in several forms. This rapid iteration capability is useful in model development and design efforts.

The example concerns the flight of a small single-stage rocket which is fired vertically. The weight of the rocket, when empty of fuel, is 600 lbs, and initially it contains 2400 lbs of fuel giving a total launch weight of 3000 lbs. The rocket produces a constant thrust (THRUST) of 7000 lbs and burns fuel at a constant rate of 40 lbs per second. The drag force (DRAG) is proportional to the square of the rocket velocity. During fuel burn the system equation is:

Y´´ = G*(THRUST-DRAG)/W-G

where W = 3000-40*T

is the weight of the rocket and fuel at time T

DRAG = K*Y´*|Y´| (to ensure drag always opposes motion)

and K = 0.008

G = 32.17 ft/sec/sec

Y = elevation from launch pad

Initial conditions are Y=0 and Y´=0 at T=0

The above equations apply until fuel is exhausted at T=60 seconds, at which point THRUST becomes zero and W becomes constant at 600 lbs. A solution for Y against T is required up to T=100 seconds.

```
:ISIM ROCKET PROGRAM
CONSTANT K=0.008,G=32.17
CONSTANT ALGO=1 : 4TH-ORDER FIXED STEP
CONSTANT TOFF=60,TFIN=100,CINT=1
INITIAL
  Y=0;Y´=0
  THRUST=7000
DYNAMIC
  W=3000-40*T
  IF (T.LE.TOFF)GOTO 1
  THRUST=0;W=3000-40*TOFF
1 DRAG=K*Y´*ABS(=Y´)
  Y´´=G*(THRUST-DRAG)/W-G
  PLOT T,Y,0,TFIN,0,50000
  PREPARE T,Y,Y´,Y´´,W,THRUST
```

The CSSL IV code for the same problem is as follows:

Rocket Dynamics (CSSL-IV)

```
PROGRAM ROCKET
INITIAL
  CONSTANT  K=0.008,G=32.17, ...
  TOFF=60.0,TFIN=100.0,YØ=0.0,...
  DYØ=0.0,THRUST=7000.0
END $"OF INITIAL SECTION"
DYNAMIC
   CINTERVAL  CINT=1.0
DRIVATIVE  ROCK
  ALGORITHM  IALGO=5, ALGO=5
  SWITCH=TOFF-T
  W=SWIN(SWITCH,3000.0-400*TOFF,3000.0-40.0*T)
  THRUST=SWIN(SWITCH,0.0,THRUST)
  DRAG=K*DY*ABS(DY)
  D2Y=G*(THRUST-DRAG)/W-G
  DY=INTEG(D2Y,DYØ)
   Y=INTEG(DY,YØ)
END $ "OF DERIVATIVE SECTION"
   TERMT(T .GE. TFIN)
   PREPAR Y,DY,D2Y,W,THRUST
 END $ "OF DYNAMIC"
TERMINAL
END
END $ "OF PROGRAM IN CSSL-IV"
```

Last is the code for the rocket problem written in IBM's DSL/VS language.

```
TITLE ROCKET PROBLEM
* DSL/AEROSPACE EXAMPLE
INITAL
CONST G=32.17,TH=7000.0,WMIN=600.0
INCON Y0=0.0,DY0=0.0
PARAM K=0.008
      CALL SCLOCK(60.0)
DERIVATIVE
      W= 3000.0 - 40.0*TIME
      WT= FCNSW( 60.0-TIME , WMIN ,WMIN , W )
      D2Y=G*TH/WT-G-K*DY*ABS(DY)*G/WT
NOSORT
      IF(TIME .GE. 60.0) TH=0.0
SORT
      DY=INTGRL(DY0,D2Y)
      Y=INTGRL(Y0,DY)
TERMINAL
PREPAR Y,DY,D2Y
CONTROL FINTIM=100.0,DELT=0.02,DELPLT=0.04
RANGE Y,DY,D2Y,WT
GRAPH (DE=TEK618) TIME,Y,   D2Y
GRAPH (DE=TEK618) TIME,Y,DY
GRAPH (DE=TEK618) TIME,  DY,D2Y
PRINT 5.0,Y,DY,D2Y
END
STOP
```

References:

(1) CSSL-IV Users Guide and Reference Manual, Simulation Services, Chatsworth, CA (213) 998-7824
(2) Dynamic Simulation Language/VS Language Reference Manual (SH20-6288-0), IBM Corporation, GPD, San Jose, CA (408) 256-4254
(3) Interactive Simulation Language User Manual, Crosbie, Hay & Associates, Chico, CA. 95927 894-8255